// Reference: https://www.rcpano.net/2021/05/25/how-to-make-remote-control-for-model-car-boat-diy-3-channel-rc/#google_vignette
// Libraries (make sure they are all properly installed)
#include <SPI.h>
#include <nRF24L01.h>
#include <RF24.h>
#include <Servo.h>
#define motor 5
// D5 to motor MOSFET gate
#define servo 3
// D3 to servo
#define peltierMode 2
// D2 to relay that controls heat/cool mode
#define peltierPower 6
// D6 to Peltier MOSFET gate
#define ledCold 7
// D7 cool mode LED
#define ledHot 8
// D8 heat mode LED
#define buzzerPin A0  
// A0 for buzzer 

int ch_width_1 = 0;
int ch_width_2 = 0;
Servo ch1;

struct Signal {
byte steering;
byte throttle;
byte button; //new
};

Signal data;
const uint64_t pipeIn = 0xE9E8F0F0E2;
// Team 1: 0xE9E8F0F0E1
// Team 2: 0xE9E8F0F0E2
// Team 3: 0xE9E8F0F0E3
// Team 4: 0xE9E8F0F0E4

RF24 radio(9, 10); // CE,CSN pins

int mode = 0; // 0 = off, 1 = cool, 2 = heat
bool lastButtonState = false;

void ResetData()
{
// Define the initial value of each data input. The middle position for Potentiometers.
data.steering = 127; // Center
data.throttle = 0; // Motor Stop
data.button = 0; // Joystick Button
}

void setup()
{
  //Set the pins for each PWM signal
  ch1.attach(servo);
  pinMode(motor, OUTPUT);

  pinMode(peltierMode, OUTPUT);
  pinMode(peltierPower, OUTPUT);
  pinMode(ledCold, OUTPUT);
  pinMode(ledHot, OUTPUT);
  pinMode(buzzerPin, OUTPUT); // Configure A0 as output for buzzer

  digitalWrite(peltierMode, LOW);
  analogWrite(peltierPower, 0);
  digitalWrite(ledCold, LOW);
  digitalWrite(ledHot, LOW);
  noTone(buzzerPin); // Ensure passive buzzer is off initially

  //Configure the NRF24 module
  ResetData();
  radio.begin();
  radio.openReadingPipe(1,pipeIn);
  radio.setAutoAck(false);
  radio.setDataRate(RF24_250KBPS);
  radio.setPALevel(RF24_PA_MAX);
  radio.startListening(); //start the radio comunication for receiver
}

unsigned long lastRecvTime = 0;

void recvData()
{
  while ( radio.available() ) {
  radio.read(&data, sizeof(Signal));
  lastRecvTime = millis();   // receive the data 
  }
}

void updatePeltierMode() {
    bool buttonPressed = data.button == 1;
    if (buttonPressed && !lastButtonState) {
        mode = (mode + 1) % 3;
        switch (mode) {
            case 0: // Off
                analogWrite(peltierPower, 0); // Peltier off
                digitalWrite(peltierMode, LOW); // relay off
                digitalWrite(ledCold, LOW);
                digitalWrite(ledHot, LOW);
                break;
            case 1: // Cool Mode
                analogWrite(peltierPower, 127); // Peltier on 50%
                digitalWrite(peltierMode, LOW); // Cool mode selected
                digitalWrite(ledCold, HIGH);
                digitalWrite(ledHot, LOW);
                break;
            case 2: // Heat Mode
                analogWrite(peltierPower, 127); // Peltier on 50%
                digitalWrite(peltierMode, HIGH); // Heat mode selected
                digitalWrite(ledCold, LOW);
                digitalWrite(ledHot, HIGH);
                break;
        }
    }
    lastButtonState = buttonPressed;
}

void updateBuzzer() {
  static unsigned long lastToggle = 0;
  const unsigned long beepInterval = 200; // Beep every 200ms for noticeability
  if (data.steering < 117 || data.steering > 137) {
    if (millis() - lastToggle >= beepInterval) {
      if (digitalRead(buzzerPin) == LOW) { // Check if tone is off
        tone(buzzerPin, 2500); // 2.5kHz tone for passive buzzer
      } else {
        noTone(buzzerPin); // Stop tone
      }
      lastToggle = millis();
    }
  } else {
    noTone(buzzerPin); // Ensure buzzer is off when centered
  }
}

void loop()
{
  recvData();
  unsigned long now = millis();
  if ( now - lastRecvTime > 1000 ) {
    ResetData(); // Signal lost. Reset data.
  }

ch_width_1 = map(data.steering, 15, 240, 500, 2500);     // pin D3 (PWM signal)
//ch_width_1 = map(data.steering, 15, 240, 1000, 2000);     // for some models of servo
ch_width_2 = map(data.throttle, 140, 255, 0, 255); // Only brighten when throttle > 140
ch_width_2 = constrain(ch_width_2, 0, 255);        // Ensure valid PWM range
   
//Write the PWM signal.
ch1.writeMicroseconds(ch_width_1);
analogWrite(motor, ch_width_2);

updatePeltierMode();
updateBuzzer(); // Update buzzer state based on steering
}

